AULA 13: Variáveis Compostas Homogêneas
Matrizes

UNIVERSIDADE FEDERAL DE PELOTAS
CENTRO DE DESENVOLVIMENTO TECNOLÓGICO
Professor: Guilherme Tomaschewski Netto
guilherme.netto@inf.ufpel.edu.br

Matrizes

- Matrizes são variáveis compostas homogêneas bidimensionais que necessitam de dois índices para individualizar um elemento do conjunto.
- O primeiro índice representa as linhas e o segundo as colunas.
- As matrizes bidimensionais possuem representação direta relacionadas a tabelas.

Matrizes

- Variáveis Indexadas Bidimensionais ou Matrizes
  - São variáveis referenciadas por dois índices, cada qual iniciando em 1.



Matrizes

- A sintaxe para declaração é:

```
Declare <id> (<linhas>,<colunas>) <tipo>
```

- Exemplos:

```
Declare valores(2,3) Numérico
Declare nomes(5,2) Literal
```

  - As declarações acima correspondem à declaração de 16 variáveis:

    valores(1,1), valores(1,2), valores(1,3),
    valores(2,1), valores(2,2), valores(2,3),
    nomes(1,1), nomes(1,2), nomes(2,1), nomes(2,2), nomes(3,1),
    nomes(3,2), nomes(4,1), nomes(4,2), nomes(5,1) e nomes(5,2),

Matrizes

- Declarar uma matriz M, de 3 linhas por 3 colunas, constituída de elementos numéricos

```
Declare M(3,3) Numérico
```

- fará com que passe a existir uma estrutura de dados agrupada denominada M, com 3x3=9 elementos inteiros, endereçáveis por um par de índices, com o primeiro indicando a linha e o outro, a coluna

Matrizes

```
Declare M(3,3) Numérico
```

$$
M = \begin{array}{|c|c|c|}
\hline
m_{11} & m_{12} & m_{13} \\
\hline
m_{21} & m_{22} & m_{23} \\
\hline
m_{31} & m_{32} & m_{33} \\
\hline
\end{array}
$$

Matrizes

- A sintaxe para atribuição é:
  <identificador>(<pos1>,<pos2>):= <exp>

- Exemplos:
  - nomes(1,1) := "João"
  - nomes(2,1) := "Maria"
  - valores(1,1) := 35
  - i := 4
  - j := 2
  - valores(i+1,j) := 45

Matrizes

- M(1,1) := 35
- ...

M =

35	10	45
-1	0	7
6	38	50

Matrizes

- Para fazer referência ou selecionar um determinado elemento da matriz usa-se dois índices:
  - um representa a linha e outro a coluna da matriz

M(linha, coluna)

M(1,1)

M(2,3)

M =

35	10	45
-1	0	7
6	38	50

Preenchendo uma matriz

- Preencher uma matriz significa atribuir valores a todas as suas posições. Para isso é necessário identificar todas as suas posições. Isto exige a utilização de um índice para cada dimensão da matriz.

- Assim como nos vetores, deve-se implementar um mecanismo que controle o valor dos índices da matriz. Pode-se, por exemplo, utilizar estruturas de repetição Para-até-faça aninhadas para garantir o acesso a cada posição das linhas e colunas da matriz.

Preenchendo uma matriz

- Preencher uma matriz bidimensional com duas linhas e cinco colunas.
- A variável i que representa as linhas deve variar dentro do intervalo de 1 a 2, ou seja, exatamente nas linhas.
- Para cada valor de i, a variável j varia de 1 a 5, ou seja, as cinco colunas que cada linha possui.

Preenchendo uma matriz

```
Algoritmo
   Declare X(2,5), i, j Numerico
   Para i de 1 até 2 faça
         Para j de 1 até 5 faça
                  Escreva “Digite o n° da linha”, i, “e coluna”, j, \n
                  Leia X(i,j)
         fim_para
   fim_para
fim_algoritmo
```

Preenchendo uma matriz - Simulação

Linha	Coluna	Saída	Entrada
i	j		
1	1	Digite o nº da linha 1 e coluna 1	12
	2	Digite o nº da linha 1 e coluna 2	9
	3	Digite o nº da linha 1 e coluna 3	3
	4	Digite o nº da linha 1 e coluna 4	7
	5	Digite o nº da linha 1 e coluna 5	-23
2	1	Digite o nº da linha 2 e coluna 1	15
	2	Digite o nº da linha 2 e coluna 2	4
	3	Digite o nº da linha 2 e coluna 3	2
	4	Digite o nº da linha 2 e coluna 4	34
	5	Digite o nº da linha 2 e coluna 5	-4

Preenchendo uma matriz - Simulação

X		1	2	3	4	5
	1	12	9	3	7	-23
	2	15	4	2	34	-4

Mostrando os elementos da matriz

- Mostrar os valores contidos em uma matriz também implica na utilização dos seus índices para a identificação das posições.

- Pode-se, por exemplo, também utilizar estruturas de repetição Para-até-faça aninhadas para garantir o acesso a cada posição das linhas e colunas da matriz.

Mostrando os elementos da matriz

```
Algoritmo
   Declare X(2,5), i, j Numerico
   Para i de 1 até 2 faça
         Para j de 1 até 5 faça
                  Escreva “Digite o nº da linha”, i, “e coluna”, j, \n
                  Leia X(i,j)
         fim_para
   fim_para
   Para i de 1 até 2 faça
         Para j de 1 até 5 faça
                  Escreva “Valor  da linha”, i, “e  coluna”, j, \n
                  Escreva X(i,j)
         fim_para
   fim_para
fim_algoritmo
```

Matriz quadrada - relações

- Uma matriz quadrada possui o mesmo número de linhas e colunas
- Algumas relações são importantes para determinar a posição dos elementos de uma matriz quadrada
- Matriz quadrada nxn

$$\begin{pmatrix} a_{11} & a_{12} & a_{13} & \cdots & a_{1n} \\ a_{21} & a_{22} & a_{23} & \cdots & a_{2n} \\ a_{31} & a_{32} & a_{33} & \cdots & a_{3n} \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ a_{n1} & a_{n2} & a_{n3} & \cdots & a_{nn} \end{pmatrix}$$

Matriz quadrada - relações

- Sendo i e j os índices dos elementos da matriz:
  - Diagonal principal: $i = j$
  - Diagonal secundária : $i + j = n + 1$
  - Abaixo da diagonal principal: $i > j$
  - Acima da diagonal principal: $i < j$
  - Acima da diagonal secundária: $i + j < n + 1$
  - Abaixo da diagonal secundária: $i + j > n + 1$

Exemplo

- Faça um algoritmo que lê uma matriz M(10,10) e calcule e escreva a soma da diagonal principal.

Exemplo - Solução

```
Algoritmo
   Declare M(10,10), i, j, SomaDPrin Numerico
   SomaDPrin:=0
   Para i de 1 até 10 faça
         Para j de 1 até 10 faça
                  Escreva "Digite o nº da linha", i, "e coluna", j, \n
                  Leia M(i,j)
                  Se i=j
                           então SomaDPrin:=SomaDPrin+M(i,j)
                  fim_se
         fim_para
   fim_para
   Escreva "Soma da diagonal principal:", SomaDPrin, \n
fim_algoritmo
```

Exercício 1

- Faça um algoritmo que para uma matriz M(10,10) atribua 1 aos elementos da diagonal principal e 0 aos demais.

Exercício 1 - Solução

```
Algoritmo
   Declare M(10,10), i, j Numerico
   Para i de 1 até 10 faça
         Para j de 1 até 10 faça
                  Se i=j
                           então M(i,j):=1
                           senão M(i,j):=0
                  fim_se
         fim_para
   fim_para
fim_algoritmo
```

Exercício 2

- Algoritmo que lê uma matriz V(3,3) e calcula e escreva as somas:
  a) da linha 3 de V;
  b) da coluna 2 de V;
  c) da diagonal principal;
  d) da diagonal secundária; e
  e) de todos os elementos da matriz

Exercício 2 - Solução

```
Algoritmo
  Declare V(3,3), i, j, l3, c2, p, s, t Numerico
  l3 := 0
  c2 := 0
  p := 0
  s := 0
  t := 0
  Para i de 1 até 3 faça
     Para j de 1 até 3 faça
        Escreva "Digite o nº da linha", i, "e coluna", j, \n
        Leia V(i,j)
        Se i = 3
          então l3 := l3 + V(i,j)
        fim_se
        Se j = 2
          então c2 := c2 + V(i,j)
        fim_se
        Se i = j
          então p := p + V(i,j)
        fim_se
        Se i + j = 4 /*n+1=3+1*/
          então s := s + V(i,j)
        fim_se
        t := t + V(i,j)
     fim_para
  fim_para
  Escreva 'Total linha 3: ', l3
  Escreva 'Total coluna 2: ', c2
  Escreva 'Total diag. princ.: ', p
  Escreva 'Total diag. sec.: ', s
  Escreva 'Total matriz: ', t
fim_algoritmo
```

Exercício 3

- Escrever um algoritmo para armazenar valores numéricos em uma matriz 5x6. A seguir, calcular e escrever a média dos valores contidos na matriz. Finalmente, escrever o conteúdo da matriz.

Exercício 3 - Solução

```
Algoritmo
    Declare M(5,6), i, j, S Numérico
    S:=0
    Para i de 1 até 5 faça
            Para j de 1 até 6 faça
                    Escreva "Digite o nº da linha", i, "e coluna", j, \n
                    Leia M(i,j)
                    S:=S+M(i,j)
            fim_para
    fim_para
    Escreva 'Média da matriz: ', S/30,\n
    Para i de 1 até 5 faça
            Para j de 1 até 6 faça
                    Escreva "Valor  da linha", i, "e  coluna", j, \n
                    Escreva M(i,j)
            fim_para
    fim_para
fim_algoritmo
```